
講義の前半は割とマメに Mathematica の例を示しているけれど、後半はさぼっているので、
今年度こそ…

1 四則など簡単な演算
虚数単位は I (大文字) で表す。絶対値 (absolute value) は Abs[], 偏角 (argument) の主値

は Arg[], 共役複素数 (complex conjugate) は Conjugate[] で計算できる。
問 11 (1) の検算に使ってみる。� �
a=1+I

b=2+3I

a+b

a-b

a b

a/b

Abs[a]

Conjugate[a]

Arg[a]

a^4� �
2 平方根の計算
c = a+ ib (a, b ∈ R) が与えられた時に、z2 = c の解を z = x+ iy (x, y ∈ R) の形で求める
ことが出来る。
(複素数の平方根が、実数の

√
で表現できる、という定理に基づく。)

(x+ iy)2 = a+ ib より連立方程式

x2 − y2 = a, 2xy = b

1http://nalab.mind.meiji.ac.jp/~mk/lecture/complex-function-2015/toi1.pdf

が得られるので、その実数の範囲の解を求めれば良い。

1 + i の平方根を求めてみよう。� �
a=1

b=1

sol=Solve[{x^2-y^2==a,2 x y==b},{x,y},Reals]

FullSimplify[sol]� �
以上は授業で説明したやり方に沿って Mathematica に仕事をさせるものだが、z の方程式
のまま解かせることも出来る (Mathematica が内部で何をしているのかは謎だけど)。� �
sol=Solve[z^2 == 1 + I, z]

sol2=ComplexExpand[sol]

ToRadicals[sol2]� �
最初に z2 = 1 + i を解かせると z = ±

√
1 + i となるが、ComplexExpand[] で実部・虚

部に展開させると、z = ± 4
√
2
(
cos

π

8
+ i sin

π

8

)
となり、ToRadicals[] で処理すると、z =

±
(√

2+
√
2

23/4
+ i

√
2−

√
2

23/4

)
となる。

3 実部虚部への分解
Ω ⊂ C, f : Ω → C とするとき、

u(x, y) := Re f(x+ iy), v(x, y) := Im f(x+ iy) ((x, y) ∈ Ω̃ :=
{
(x, y) ∈ R2 | x+ iy ∈ Ω

}
)

で定めた u, v (複素関数の実部・虚部) が必要になる場合がある。

これらを求めるには、既に紹介した ComplexExpand[] を用いると良い。� �
f[z_]:=z^3

ComplexExpand[f[x+I y]]

g[z_]:=Cos[z]

ComplexExpand[g[x+I y]]� �
4 (有限項) Taylor 展開
有限項までで打ち切ったものしか計算出来ないが、Series[式, {変数, c, n}] で、c のま

わりの Taylor 展開を第 n項まで計算出来る。

� �
Series[Cos[z],{z,0,10}]

Series[1/(2+z),{z,0,10}]� �
5 級数
Mathematica の Sum[] は級数の和を計算してくれる。思いの外、強力である。� �
Sum[(-1)^n z^n/(2n)!,{n,0,Infinity}]� �
に対して Cos[z] という答を返す。
Taylor 展開、Laurent 展開は、ある程度、自分で計算する必要があるようだが、その結果を

Sum[] に与えることで検算が可能である。

6 部分分数分解
有理式の部分分数への分解が必要になる場合があるが、Apart[] で計算できる。� �
Apart[(z^3-3z^2-z+5)/(z^2-5z+6)]� �

7 応用: ある計算問題 (有理関数の Taylor 展開を求める) の
答の検算

f(z) =
z3 − 3z2 − z + 5

z2 − 5z + 6
.

の 0 の周りの Taylor 展開を求めよ、という問題。
人手で解く場合は、f(z)を部分分数に分解する。そのためには、z3−3z2−z+5を z2−5z+6

で割りたくなる。� �
A=z^3-3z^2-z+5

B=z^2-5z+6

f=A/B

q=PolynomialQuotient[A,B,z]

r=PolynomialRemainder[A,B,z]� �

これから商 q = z+2,余り r = 3z−7が求まる (商は quotient,余りは remainder. polynomial

は多項式という意味)。ゆえに

f(z) =
(z + 2)(z2 − 5z + 6) + 3z − 7

z2 − 5z + 6
= z + 2 +

3z − 7

(z − 2)(z − 3)
.

この右辺を部分分数分解しても良いが、そもそも Mathematica にやらせるのならば、最初か
ら f(z) の部分分数分解を指示してもよい。� �
Apart[r/B]

Apart[f]� �
それぞれ

2

−3 + z
+

1

−2 + z
, 2 +

2

−3 + z
+

1

−2 + z
+ z となる。

結局

f(z) = z + 2 +
1

z − 2
+

2

z − 3
.

Series[f,{z,0,10}] とすると、0 の周りの Taylor 展開を 10 次の項まで求めることが出
来る。

f(z) =
5

6
+

19z

36
− 43

216
z2 − 113

1296
z3 − 307

7776
z4 − · · · (途中省略)− · · · − 181243

362797056
z10 +O(z11)

が得られる。series は級数という意味の英単語である。
残念ながら、Mathematica で Taylor 展開の一般項を求めるための簡単な手段は用意されて
いないようである。
人手で計算して、

f(z) =
5

6
+

19

36
z −

∞∑
n=2

(
1

2n+1
+

2

3n+1

)
zn

が求まるが、この結果の検算は可能で� �
5/6+19z/36-Sum[(1/2^(n+1)+2/3^(n+1))z^n,{n,2,Infinity}]

Simplify[%]� �
とすると、

5− z − 3z2 + z3

6− 5z + z2

が得られる。無事、f(z) と一致したので、ほっと一息。

8 複素対数関数を描く
2変数 (x, y) の関数としての ImLog(x+ iy), Re Log(x+ iy) のグラフを描いてみよう。
それぞれ Arg(x + iy), log

√
x2 + y2 であるから、コンピューターで図示しなくても分から

なくはないが (図示しなくても分かるけれど)、やってみることを勧める。

Plot3D[] や ContourPlot[] では、描画範囲を x座標と y座標の範囲で指定するので、変
数は x+I y と書くと良い (小さなノウハウ)。� �
Plot3D[Boole[x^2+y^2<4] Im[Log[x+I y]],{x,-2,2},{y,-2,2}]

Plot3D[Boole[x^2+y^2<4] Re[Log[x+I y]],{x,-2,2},{y,-2,2}]� �
(Boole[] はなくても良い。描画範囲を円 x2 + y2 = 4 内に制限しているだけで、あまり深

い意味はない。個人的な趣味です。)

Mathematica で描いたグラフは、マウスでつかんでグリグリ動かせる。ぜひやってみるこ
と (静止画を見るだけだと今ひとつ分かりにくい)。
Plot3D[] の代わりに ContourPlot[] を用いると、レベル表示 (≒等高線描画 ) 出来る。� �
ContourPlot[Im[Log[x+I y]], {x,-2,2}, {y,-2,2}, Contours->Table[x,{x,-Pi,Pi,Pi/8}]]

ContourPlot[Re[Log[x+I y]],{x,-2,2},{y,-2,2}]� �
9 Abel の連続性定理に現れる収束範囲

冪級数
∞∑
n=0

an(z− c)n が収束円周上のある点 z0 で収束するならば、その冪級数は “Stolzの

角領域” で一様収束するので、和はそこで連続な関数である、というのが Abel の連続性定理
で、それにより、

π

4
= 1− 1

3
+

1

5
− · · ·+ (−1)n−1

2n− 1
+ · · ·

などの有名な結果が証明できる。
例えば c = 0, z0 = R の場合、

ΩK :=

{
z ∈ C

∣∣∣∣ |z| < R,
|1− z/R|
1− |z|/R

< K

}
であるが、これは一体どういう形なのだろうか？

� �
stolz[K_, R_] :=

Block[{g1, g2},

g1 = ContourPlot[x^2 + y^2 == R^2, {x, -2 R, 2 R}, {y, -2 R, 2 R}];

g2 = RegionPlot[

x^2 + y^2 < R^2 &&

Abs[1 - (x + I y)/R]/(1 - Abs[x + I y]/R) <= K, {x, -2 R,

2 R}, {y, -2 R, 2 R}]; Show[g1, g2]

]

R=1

Manipulate[stolz[K,R],{K,1,10,0.2}]� �

図 1: K を大きくすると膨れていきます

筆者は、Mathematica を使うまで、ΩK がどういう形をしているか、実は良く分かっていな
かった (そんなに難しくもないけれど、ちょっと考えて分かるものでもなくて、何となく気に
なってはいたけれど、放置していました。)。

10 曲線の連続的な変形 (ホモトピー)

円を楕円に変形する。� �
phi0[t_]:={Cos[t],Sin[t]};

phi1[t_]:={3Cos[t],2Sin[t]};

F[t_,u_]:=(1-u)phi0[t]+u*phi1[t];

Manipulate[ParametricPlot[{phi0[t], phi1[t], F[t, u]}, {t, 0, 2 Pi},

PlotRange -> {{-3, 3}, {-3, 3}}], {u, 0, 1}]� �
phi1[t ]:={1/2,0} とすると、定数曲線で、像は {(1/2, 0)} であり、円周 x2 + y2 = 1を 1

点 (1/2, 0) に変形することになる。
以上は複素数を使っていないので、書き換えてみる。� �
Clear[phi0,phi1,F]

phi0[t_]:=Exp[I t];

phi1[t_]:=3Cos[t]+2 I Sin[t];

F[t_,u_]:=(1-u)phi0[t]+u*phi1[t];

z2xy[z_]:={Re[z],Im[z]}

Manipulate[ParametricPlot[{z2xy[phi0[t]],z2xy[phi1[t]],z2xy[F[t,u]]},

{t, 0, 2 Pi}, PlotRange -> {{-3, 3}, {-3, 3}}],

{u, 0, 1}]� �

